Transportation in GAMS

Mathematical Representation

I tried to modify the model of transport in GAMS. As an example (page 68 for chapter 4), I use the potential fishing industry with canneries in Seattle, San Diego, Los Angeles and Portland and markets in New York, Chicago and Topeka. A number of these four potential fishing industries are chosen instead of open two predetermined industries. In this model I not only seek to find the pattern of shipments from the canneries to markets which will have the least transportation cost but also consider the investment costs when open the industry while satisfying the fixed demand at the markets without shipping more from any cannery than its' capacity.

The model is stated mathematically as:

For the sets

I plants = {Seattle, San Diego, LosAngeless, Portland}

J markets= {New York, Chicago, Topeka}

Find

x_{ij} shipments from plant i to market j

y_i binary variable, whether plant i is open, if i is open the value is one, otherwise zero

to minimize total costs including the transportation cost and the investment costs

(1) $$z = \sum_{i \in I} \sum_{j \in J} c_{ij} x_{ij} + \sum_{i \in I} f_i y_i$$

Where

c_{ij} Transportation cost from plant i to market j per unit shipped

f_i Fixed investment cost when open plant i

The function (1) is minimized subject to the constraints that no more be shipped from each plant than its capacity

(2) $$\sum_{j \in J} x_{ij} \le a_i \qquad i \in I$$

Where

a_i The capacity of plant i

And that no less be shipped to each market than its demand

(3) $\sum_{i \in I} x_{ij} \geq b_j \qquad j \in J$

Where

b_j The demand at market j

Not all of the plants are going to be open, only some of the plants are going to be open

(4) $\sum_{i \in I} y_i = N$

Where

N Number of plants that are going to be opened

While nothing will be shipped from a plant if this is not open,

(5) $\sum_{j \in J} x_{ij} \leq y_i \cdot M$

Where

M is a big positive number

While requiring that all the shipments be non-negative

(6) $x_{ij} \geq 0 \qquad i \in I \quad j \in J$

GAMS model

In this model, the number of plants I am going to open is 2. Where N=2.

```
 SETS
     I   canning plants   / SEATTLE, SAN-DIEGO, LosAngeless, Portland /
     J   markets          / NEW-YORK, CHICAGO, TOPEKA /;

 PARAMETERS

* I wanna open two of the four potential plants instead of just consider two plants in total

     A(I)  capacity of plant i in cases
       /   SEATTLE         350
           SAN-DIEGO       600
           LosAngeless     500
           Portland        580   /

     B(J)  demand at market j in cases
       /   NEW-YORK    325
```

```
          CHICAGO     300
          TOPEKA      275  /
*not only consider the transportation cost but also consider the investment cost of facilities

       FC(I) fixed cost of plant i in cases  in thousands of dollars
       /   SEATTLE          6000
           SAN-DIEGO        12000
           LosAngeless               7000
           Portland                  9200 /;

 TABLE D(I,J)  distance in thousands of miles

              NEW-YORK      CHICAGO     TOPEKA
    SEATTLE        2.5          1.7          1.8
    SAN-DIEGO      2.5          1.8          1.4
    LosAngeless    2.55         1.8          1.3
    Portland       2.4          1.75         1.3 ;

 SCALAR
       F  freight in dollars per case per thousand miles  /90/
       N  number of plants open /2/
       M  a big number /10000000000/;

 PARAMETER C(I,J)  transport cost in thousands of dollars per case ;

         C(I,J) = F * D(I,J) / 1000 ;

 VARIABLES
     X(I,J)  shipment quantities in cases
     Z       total transportation costs in thousands of dollars
     Y(I)    plant i is open value is one otherwise zero

 POSITIVE VARIABLE X ;
 BINARY VARIABLE Y;

 EQUATIONS
     TTCOST            define objective function shipment costs plue fixed costs
     SUPPLY(I)         observe supply limit at plant i
     DEMAND(J)         satisfy demand at market j
     OPEN              number of plants going to open
     CONSTRAINT(I)  if a plant if not open its quantities is zero;

 TTCOST ..         Z  =E=  SUM((I,J), C(I,J)*X(I,J))+ sum(I, FC(I)*Y(I)) ;
 SUPPLY(I) ..      SUM(J, X(I,J))  =L=  A(I) ;
 DEMAND(J) ..      SUM(I, X(I,J))  =G=  B(J) ;
 OPEN..            SUM(I,Y(I))     =E=  N   ;

*if a plant if not open, its quantities is zero
 CONSTRAINT(I)..   SUM(J, X(I,J)) =L= Y(I)*M ;

 MODEL TRANSPORT /ALL/ ;
 SOLVE TRANSPORT USING MIP MINIMIZING Z ;
 DISPLAY X.L,Y.L;
```

Results

```
S O L V E     S U M M A R Y

    MODEL   TRANSPORT          OBJECTIVE  Z
    TYPE    MIP                DIRECTION  MINIMIZE
    SOLVER  CPLEX              FROM LINE  88

**** SOLVER STATUS     1 NORMAL COMPLETION
**** MODEL STATUS      1 OPTIMAL
**** OBJECTIVE VALUE          15348.4550

----     90 VARIABLE X.L  shipment quantities in cases

                NEW-YORK     CHICAGO      TOPEKA

SEATTLE         20.000       300.000
Portland        305.000                   275.000


----     90 VARIABLE Y.L  plant i is open value is one otherwise zero

SEATTLE  1.000,    Portland 1.000
```

The results show that if 2 plants of the potential four plants are going to be opened. They are Seattle and Portland given the parameter specified in GAMS model. And the shipment from Seattle to New York would be 20 to Chicago would be 300 while shipments from Portland to New York would be 305 and to Topeka would be 275. The total cost (including transportation costs and fixed investment cost) is 15348.4550 K dollars.